# Maintainer: Jan Alexander Steffens (heftig) <heftig@archlinux.org>
# Contributor: Jan de Groot <jgc@archlinux.org>

pkgbase=eog
pkgname=(eog eog-docs)
pkgver=44.1
pkgrel=1
pkgdesc="Eye of Gnome: An image viewing and cataloging program"
url="https://wiki.gnome.org/Apps/EyeOfGnome"
arch=(x86_64)
license=(GPL)
depends=(
  exempi
  gnome-desktop
  lcms2
  libexif
  libhandy
  libpeas
  librsvg
  webp-pixbuf-loader
)
makedepends=(
  gi-docgen
  git
  gobject-introspection
  meson
  systemd
  yelp-tools
)
_commit=d989d8ac5e60b8c9c429cbe67f3c79b26f22d75b  # tags/44.1^0
source=("git+https://gitlab.gnome.org/GNOME/eog.git#commit=$_commit")
b2sums=('SKIP')

pkgver() {
  cd eog
  git describe --tags | sed 's/[^-]*-g/r&/;s/-/+/g'
}

build() {
  local meson_options=(
    -D gtk_doc=true
    -D libportal=false
  )

  arch-meson eog build "${meson_options[@]}"
  meson compile -C build
}

check() {
  meson test -C build --print-errorlogs
}

package_eog() {
  optdepends=('eog-plugins: Additional features')
  groups=(gnome)

  meson install -C build --destdir "$pkgdir"

  mkdir -p doc/usr/share
  mv {"$pkgdir",doc}/usr/share/gtk-doc
}

package_eog-docs() {
  pkgdesc+=" (documentation)"
  depends=()

  mv doc/* "$pkgdir"
}

# vim:set sw=2 sts=-1 et:
